

Exploiter Pspice pour simuler des filtres numériques

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet :

[page d'accueil](#)

d'autres pdf, sur différents sujets :

[liste des PDE](#)

La synthèse des filtres numériques (ou traitements numériques) est un très vaste sujet d'étude. Il est proposé dans cet article, des exemples simples de complexité croissante. Pour vérifier le comportement fréquentiel et/ou temporel, on peut utiliser Pspice.

1) Exemple 1 :

Cahier des charges :

On dispose d'un traitement analogique, dont la transmittance en Laplace est : $T(p) = \frac{1}{(1 + \tau p)}$.

On cherche à déterminer une transmittance $T(z)$ équivalent, donnant une réponse harmonique *similaire*.

1.a) Synthèse choisie

On choisit ici la méthode du TRAPEZE, appelée aussi de Tustin (voir [annexe 1](#)) :

La variable p est remplacée par $\frac{2(z-1)}{T(z+1)}$. On aboutit directement à $T(z) = \frac{a - b z^{-1}}{1 - c z^{-1}}$

avec :

$a = \frac{T}{T + 2\tau}$	$b = -\frac{T}{T + 2\tau}$	$c = \frac{2\tau - T}{T + 2\tau}$
---------------------------	----------------------------	-----------------------------------

On a $T(z) = \frac{a - b z^{-1}}{1 - c z^{-1}} = \frac{S(z)}{E(z)}$.

L'équation aux différences s'obtient par le produit en croix :

Et en prenant l'original terme à terme :

Il vient :

$$(a - b z^{-1}) E(z) = (1 - c z^{-1}) S(z).$$

$$a e_{(nT)} - b e_{(n-1)T} = S_{(nT)} - c S_{(n-1)T}$$

$$S_{(nT)} = a e_{(nT)} - b e_{(n-1)T} + c S_{(n-1)T}$$

1.b) Application numérique

- Choisissons $\tau = 50 \mu\text{s}$, ce qui donne une réponse harmonique analogique $20 \log |T(jf)|$ formée d'un plateau à 0 dB et d'une fréquence de coupure à -3 dB située à $\frac{1}{2\pi\tau} = 3,183 \text{ kHz}$.

- On fixe la fréquence d'échantillonnage à 62,5 kHz (par exemple, obtenue par le timer0 d'une carte Arduino Uno, avec la fréquence d'entrée la plus élevée et le diviseur le plus petit : `setPwmFrequency(6, 1)`). On a alors **T = 16 μs** .


$$a = \frac{16}{16 + 2 \times 50} = 0,13793... \quad b = -\frac{16}{16 + 2 \times 50} = -0,13793... \quad c = \frac{2 \times 50 - 16}{16 + 2 \times 50} = 0,72413...$$

1.c) Schéma correspondant fichier exemple_1.cir



T : retard d'une période = 16 μs.

On ajoute une fonction $\frac{1}{1 + \tau p}$ qui sert de sortie témoin.

L'entrée n'est pas échantillonnée, car nous allons faire une simulation AC.



1.d) Simulation



réponse harmonique représentée sur 2 échelles des fréquences: log et lin

Interprétation

Plot supérieur :

L'échelle semi log est usuelle pour représenter des réponses harmoniques : sur la réponse « analogique » on reconnaît le plateau à 0 dB, puis une asymptote à -20 dB/décade.

La fréquence de coupure est quasiment à la valeur attendue 3,183 kHz, le très faible écart étant lié à la résolution fréquentielle de l'analyse AC.

Cutoff_Lowpass_3dB(V(9))	3.17555k
Cutoff_Lowpass_3dB(V(8))	3.14909k

La réponse « numérique », c'est à dire l'analyse AC de l'équation aux différences, est confondue avec la version analogique en basse fréquence. Sa fréquence de coupure est à 3,149 kHz, car sa réponse est plus « plongeante ». La fonction de transfert passe par 0 (d'où la descente vers -∞ en dB) à Fech/2, soit 31,25 kHz. Puis, la réponse est reproduite « symétriquement », ce qui est peu visible sur une échelle log.

Plot inférieur :

C'est exactement la même information, mais présentée cette fois sur une échelle linéaire en fréquence. C'est, bien sûr, la courbe numérique qui est mise en valeur, surtout par ce choix d'avoir 1 graduation = 31,25 kHz. On reconnaît aisément les duplications de motifs aux multiples de la fréquence d'échantillonnage (62,5 kHz), et la symétrie par rapport à Fech/2. D'ailleurs, il est d'usage de ne montrer que l'intervalle [0 ; Fech/2].

En conclusion, la synthèse qui a donné l'équation aux différences respecte le cahier des charges, mais dans le domaine des basses fréquences. Il est évident que les réponses analogique et numérique se séparent fatalement à l'approche de Fech/2.

Si on veut chiffrer la bande de fréquence commune, on peut proposer de 0 à 10 kHz, ce qui est visible sur les graphes (à cette échelle en ordonnées) : à 10 kHz, on a -11,074 dB pour V(8) et -10,362 dB pour V(9). On voit également sur ces graphes l'intérêt de respecter la fréquence de Shannon : il faut que les fréquences du signal soit inférieures à Fech/2. En effet, au-delà de 31,25 kHz, le filtre numérique propose une réponse harmonique d'allure opposée du filtre analogique.

2) Exemple 2 :

Cahier des charges :

on dispose d'un traitement analogique donné par : $T(p) = A \frac{1 + \alpha \tau p}{1 + \tau p}$.

On cherche à déterminer une transmittance $T(z)$ équivalente, telle que sa sortie passe par les mêmes valeurs, aux instants d'échantillonnage, que celle du traitement analogique, lorsqu'elle est soumise à une échelon.

2.a) Synthèse choisie

Il faut ici la méthode de l'invariant indiciel (voir annexe 2) :

Rappelons que la réponse indicielle du traitement analogique s'écrit : $s(t) = A [1 + (\alpha - 1) \exp(-t/\tau)] u(t)$

On déduit $S(z)$, par un formulaire de transformées en z : $S(z) = A \left[\frac{z}{z-1} + (\alpha - 1) \frac{z}{z - \exp(-T/\tau)} \right]$

L'invariant indiciel signifie : $Z\{s(kT)\} = \frac{z}{z-1} T(z)$,

c'est-à-dire : $A \left[\frac{z}{z-1} + (\alpha - 1) \frac{z}{z - \exp(-T/\tau)} \right] = \frac{z}{z-1} T(z)$

On déduit : $T(z) = A \left[1 + (\alpha - 1) \frac{z-1}{z - \exp(-T/\tau)} \right]$ qui peut s'écrire : $\frac{a z - b}{z - c}$ ou $\frac{a - b z^{-1}}{1 - c z^{-1}}$

Avec $a = A \alpha$ $b = A [\alpha + \exp(-T/\tau) - 1]$ $c = \exp(-T/\tau)$

D'où l'équation aux différences (voir 1.a) : $s_{(nT)} = a e_{(nT)} - b e_{(n-1)T} + c s_{(n-1)T}$


2.b) Application numérique

- Choisissons $A = 2$; $\alpha = 6$; $\tau = 80 \mu s$.
- On fixe la fréquence d'échantillonnage à 50 kHz. On a alors $T = 20 \mu s$.

On a : $a = 2 \times 6 = 12$ $b = 2 [6 + \exp(-20/80) - 1] = 11,5576...$ $c = \exp(-20/80) = 0,7788...$
 D'où l'équation aux différences : $s_{(nT)} = 12 e_{(nT)} - 11,5576 e_{(n-1)T} + 0,7788 s_{(n-1)T}$


2.c) Schéma correspondant fichier exemple_2.cir

Pour réaliser $e_{(nT)}$ il faut un échantillonneur. Il peut être représenté par un interrupteur parfait, ce qui simulable par Pspice par le "switch".



Le signal de commande est paramétré pour avoir un état passant de 10 ns pour une période de 20 μs . Cela permet d'être très proche du cas idéal.

2.d) Simulation



Réponses « numérique » et « analogique ». Une graduation = 20 μs, c'est à dire une période T

Interprétation

Le signal de sortie est échantillonné, par héritage de l'entrée. La sortie reste inchangée entre 2 valeurs, ce qui donne l'aspect d'un signal échantillonné bloqué.

Pour vérifier la coïncidence avec $s(t)$, on ajoute, sur le plot, l'équation $s(t) = A [1 + (\alpha - 1) \exp(-t/\tau)]$. La valeur initiale est 12 V, la valeur finale est 2 V. Et, comme on peut le voir, à chaque instant d'échantillonnage, $s(t)$ est en parfaite coïncidence avec $s(nT)$.

3) Exemple 3 :

Cahier des charges : on veut réaliser un filtre rejecteur qui élimine la fréquence f_0 .

La fonction de transfert analogique est, sous forme de Laplace : $\frac{1 + (\tau p)^2}{(1 + \tau p)^2}$.

En régime harmonique, le numérateur présente un 0 à la fréquence $1/(2 \pi \tau)$, que l'on pose à f_0 . Cette fréquence est donc éliminée. Le dénominateur contre carre la pente à + 40 dB / décade du numérateur, et ce, dès $1/(2 \pi \tau)$. La réponse harmonique présente donc un plateau à 0 dB, sauf à $f = f_0$, où $20 \log$ du module présente un « puits » vers $-\infty$.

3.a) Synthèse choisie

On choisit ici la méthode du TRAPEZE (voir **annexe 1**) : la variable p est remplacée par $\frac{2(z-1)}{T(z+1)}$.

$T(z)$ s'écrit alors : $\frac{1 + (\tau \frac{2(z-1)}{T(z+1)})^2}{(1 + \tau \frac{2(z-1)}{T(z+1)})^2}$, ce qui, après quelques lignes de calculs, donne : $\frac{a + b z^{-1} + c z^{-2}}{d + e z^{-2} + f z^{-2}}$

Avec

$a = T^2 + 4 \tau^2$
$d = T^2 + 4 \tau T + 4 \tau^2$

$b = 2 T^2 - 8 \tau^2$
$e = 2 T^2 - 8 \tau^2$

$c = T^2 + 4 \tau^2$
$f = T^2 - 4 \tau T + 4 \tau^2$

Il vient l'équation aux différences du traitement numérique :

$$s_{(nT)} = \frac{a}{d} e_{(nT)} + \frac{b}{d} e_{(n-1)T} + \frac{c}{d} e_{(n-2)T} - \frac{e}{d} s_{(n-1)T} - \frac{f}{d} s_{(n-2)T}$$

3.b) Application numérique

- Cherchons à éliminer la fréquence 100 Hz. Il vient $\tau = 1/(2\pi f_0) = 1,5915 \text{ ms}$
- On fixe la fréquence d'échantillonnage à 1 kHz. Il vient $T = 1 \text{ ms}$.


On déduit :

$$s_{(nT)} = 0,6362 e_{(nT)} + (-1,044) e_{(n-1)T} + 0,6362 e_{(n-2)T} - (-1,044) s_{(n-1)T} - 0,2724 s_{(n-2)T}$$


3.c) Schéma correspondant fichier exemple_3.cir

T : retard d'une période = 1 ms.
On ajoute une fonction qui sert de sortie témoin.

L'entrée n'est pas échantillonnée, car nous allons faire une simulation AC.



3.d) Simulation




Réponses harmonique, échelle linéaire en fréquence

Interprétation

En **analogique**, la fréquence 100 Hz est éliminée.

En **numérique**, malgré la précision des coefficients, la réjection se fait à 97 Hz. Cela est dû à l'approximation faite par la méthode du trapèze. Il y a une symétrie à $F_{ech}/2 = 500 \text{ Hz}$.

Pour obtenir exactement 100 Hz, il suffit de retoucher les coefficients : le plus simple, dans le fichier.cir, est de modifier f_0 jusqu'à ce que la simulation du traitement numérique donne une réjection à 100 Hz exactement. C'est obtenu pour $f_0 = 103,425 \text{ Hz}$.



autres paramètres pour avoir exactement $f_0 = 100 \text{ Hz}$.

Pour information, l'équation aux différences est alors :

$$s_{(nT)} = 0,62981 e_{(nT)} + (-1,01905) e_{(n-1)T} + 0,62984 e_{(n-2)T} - (-1,01905) s_{(n-1)T} - 0,25962 s_{(n-2)T}$$


4) Exemple 4 :

Cahier des charges :

on dispose d'un asservissement équipé d'un correcteur P.I.D.

H(p) est l'association processus + capteur de retour.

On appelle $\epsilon(t)$ le signal d'erreur, en sortie du comparateur.




Le **PID** est sous forme parallèle, dont le bloc K_p a glissé dans la chaîne directe, ce qui est une configuration fréquente.

Les transmittances respectives sont :

P : K_p . C'est un simple coefficient multiplicateur.

I : $\frac{K_i}{p}$ C'est la fonction intégratrice de base.


Pour information, sa réponse harmonique est rappelée ci-contre.



D : $\frac{K_d p}{1 + \tau p}$. Il s'agit d'un « pseudo dérivateur », c'est à dire d'un

dérivateur dont la bande passante n'est pas infinie, ce qui est proche de la réalité physique dans une large bande de fréquence.

Sa réponse harmonique est rappelée ci-contre.



On veut remplacer le PID analogique par son équivalent numérique.

Remarque : il est nécessaire d'intercaler des convertisseurs **analogique/numérique** et **numérique/analogique** pour câbler la boucle entièrement.



On supposera ces convertisseurs parfaits : temps de conversion négligeables devant le temps de réponse de l'asservissement, et une résolution suffisamment élevée pour négliger l'effet de discrétisation des valeurs. Par contre, il faut tenir compte d'un bloqueur associé au convertisseur Analogique / Numérique.

4.a) Synthèse choisie pour le correcteur PID.

Une façon simple de procéder est de synthétiser chaque branche, et de sommer les sorties respectives.

On choisit la transposition de Tustin. Ainsi,

➤ la branche **I** est : $K_i \frac{T(z+1)}{2(z-1)} = \frac{K_i T}{2} \frac{1+z^{-1}}{1-z^{-1}} = \frac{S_i(z)}{\epsilon(z)}$

ce qui donne une équation aux différences : $S_{i(nT)} = \frac{K_i T}{2} \epsilon_{(nT)} + \frac{K_i T}{2} \epsilon_{(n-1)T} + S_{i(n-1)T}$

➤ la branche **D** est $\frac{K_d \frac{2(z-1)}{T(z+1)}}{1 + \tau \frac{2(z-1)}{T(z+1)}} = \frac{2 K_d (1-z^{-1})}{T(1+z^{-1}) + 2\tau(1-z^{-1})} = \frac{S_d(z)}{\epsilon(z)}$

ce qui donne une équation aux différences : $S_{d(nT)} = \frac{2 K_d}{2\tau + T} \epsilon_{(nT)} - \frac{2 K_d}{2\tau + T} \epsilon_{(n-1)T} - \frac{T - 2\tau}{2\tau + T} S_{d(n-1)T}$

4.b Application numérique

La transmittance $H(p)$ est : $\frac{10}{p^3 + 4p^2 + 4p + 1}$.

Elle est du type passe bas (coupe à -3 dB vers 50 mHz). Rebouclée avec un retour unitaire, elle présente, lors d'un essai indiciel, une réponse oscillatoire amortie (premier dépassement à 54 % et temps de réponse à 5% de 20,6 secondes). L'intervention d'un correcteur est justifiée.


Un essai préalable (présenté au paragraphe 4.e) permettant un comportement satisfaisant de la boucle fermée analogique a donné les valeurs numériques suivantes au correcteur PID analogique :

$K_p = 0,55$ $K_i = 0,27 \text{ rad/s}$ $K_d = 1 \text{ s}$ $\tau = 0,05 \text{ s}$

La période d'échantillonnage est fixée à **10 ms**, soit une fréquence d'horloge de 100 Hz.

4.c) Schéma correspondant du PID numérique

Il est précédé d'un échantillonneur bloqueur. $e(t)$ est l'entrée analogique. Le signal après l'échantillonneur bloqueur est $epsn$: dans l'environnement final cela sera le signal d'erreur ϵ en numérique.



4.d) Simulation du PID

Dans un premier temps, on teste les branches indépendamment :

Test de la branche I : fichier Ki.cir

On remarque que l'équation aux différences est très similaire à celle simulée au paragraphe 2.c.


Par simplicité, on reprend son schéma et la netliste.

La sortie est sin (sortie intégrateur numérique).

On ajoute un intégrateur analogique témoin. Sa sortie est sia (sortie intégrateur analogique).

On a alors : $sin_{(nT)} = a \epsilon_{(nT)} + b \epsilon_{(n-1)T} + c sin_{(n-1)T}$

où : $a = 0,00135$ $b = 0,00135$ $c = 1$




Remarque :

pour des raisons de convergence lors de la simulation temporelle de Pspice, on s'interdit $c = 1$.

On place $c = 0,99999$.

Essai à l'échelon unitaire



Réponse à l'échelon unitaire de l'intégrateur seul

Interprétation


On place une valeur constante à l'entrée, ici 1 V. L'intégration donne une rampe.

Sortie analogique : avec la condition initiale nulle, $s(t)$ est $K_i t$ (pour $t > 0$).

En 0,2 seconde, la sortie a augmenté de $K_i \times 0,2 = 0,054$ V, ce que nous avons sur le graphe.

Sortie numérique : On dispose de la même pente, mais le signal est échantillonné bloqué. La durée de chaque échantillon est $T = 10$ ms.

Essai à la rampe unitaire



Réponse à la rampe unitaire de l'intégrateur seul

Interprétation :

Si l'entrée est une rampe unitaire, la sortie analogique a pour équation $\frac{K_i t^2}{2}$ (pour $t > 0$). Après 0,2

seconde, l'accroissement est de $\frac{0,27 \times 0,2^2}{2} = 5,4$ mV. La réponse échantillonnée suit cette parabole.

En conclusion de ces 2 simulations, on peut donc valider le bloc « I » numérique.

Test de la branche D : fichier Kd.cir

L'équation aux différences est du même type. Par simplicité, on reprend son schéma et la netliste. La sortie est sdn. On ajoute un dérivateur analogique témoin. Sa sortie est sda.





Schéma simulé pour le test du correcteur D. en rouge, les numéros de nœuds

On a alors : $sdn_{(nT)} = a \epsilon_{(nT)} + b \epsilon_{(n-1)T} + c sdn_{(n-1)T}$

avec : $a = 18,18...$ $b = -18,18...$ $c = -0,818...$

Essai indiciel



Réponse à l'échelon unitaire du pseudo dérivateur seul

Interprétation

La réponse indicielle d'une transmittance d'équation $\frac{K_d p}{1 + \tau p}$ est $\frac{K_d}{\tau} \exp(-t/\tau)$, soit ici, $20 \exp(-t/0,05)$.

C'est ce qu'on retrouve sur la réponse **analogique** : par exemple à $t = 100 \text{ ms}$, $20 \exp(-0,1/0,05) = 2,7 \text{ V}$, valeur retrouvée sur la courbe.

La version **numérique** montre cette courbe échantillonnée (à 100 Hz) et bloquée.

Dans un deuxième temps, on teste le PID isolément :

Test PID en temporel : fichier PID.cir

On choisit de placer un échelon en entrée. Pour mieux lire les courbes de simulation, on modifie, uniquement pour ce test, la valeur de K_i qui passe à 10 (valeur arbitraire).

La sortie du PID numérique est spn.

On ajoute un PID analogique témoin. Sa sortie est spa.





Schéma simulé pour le test du correcteur PID. en rouge, les noms des nœuds



Réponse à l'échelon unitaire du PID seul. Ici $K_p = 0,55$ $K_i = 10 \text{ rad/s}$ $K_a = 1 \text{ s}$ $\tau = 0,05 \text{ s}$.

Interprétation

On retrouve bien la contribution de l'impulsion et de la rampe. Plus précisément, la sortie de ce PID analogique obéit à l'équation (pour $t > 0$) : $[10 t + 20 \exp(-t/0,05) + 1] \times 0,55$.

On vérifie : à $t = 0$, l'équation aboutit à $21 \times 0,55 = 11,55$, ce qui est la valeur affichée,

à $t = 1 \text{ s}$, l'équation aboutit à $[10 + 0 + 1] \times 0,55 = 6,05$, ce qui est la valeur affichée.


La sortie numérique « colle » à cette réponse, en étant échantillonnée bloquée.

Test PID en régime harmonique:


Par curiosité scientifique, intéressons-nous à la réponse harmonique du correcteur PID.

Il suffit que l'entrée soit une source AC, et de forcer la clk à l'état haut, afin que l'échantillonneur soit passant en permanence. Le condensateur qui fait le bloqueur est sans effet.

On replace la bonne valeur à K_i .



comparaison des réponses harmoniques



Réponses en régime harmonique du PID seul analogique et numérique

Interprétation

> En basse fréquence (vers 10 mHz), c'est la réponse de l'intégrateur qui prédomine : on a bien la pente à - 20 dB / dec.

Le passage à 0 dB de l'asymptote est à la fréquence $\frac{K_p K_i}{2 \pi} = \frac{0,55 \times 0,27}{2 \pi} = 23,6$ mHz, ce qui est vérifié en simulation.

> En haute fréquence (ici vers 1 Hz), c'est le dérivateur qui prédomine. Dans la zone de fréquence avant la cassure $\frac{1}{2 \pi \tau} = 3,18$ Hz, la pente est à + 20 dB / déc. Sur l'asymptote, la traversée à 0 dB est à la

fréquence $\frac{1}{2 \pi K_p K_d} = \frac{1}{2 \pi \times 0,55} = 289$ mHz, ce qui est vérifié en simulation.

A $f \gg 3,18$ Hz, les réponses présentent un plateau à $20 \log \frac{K_p K_d}{\tau} = 20,8$ dB.

La bande de fréquence où le PID est fonctionnel est de 0 Hz à 3 Hz environ. Dans cette bande, les 2 réponses sont confondues. C'est à l'approche de Fclk = 100 Hz que les courbes diffèrent.


La réponse **numérique** est symétrique par rapport à Fclk/2 = 50 Hz. Comme à f = 0, le module tend vers l'infini (par l'action de l'intégrateur), on retrouve ce phénomène à 100 Hz, par symétrie. Ce qui explique la courbe qui diverge à 100 Hz exactement.

4.e) Simulation du système complet


On place le processus H(p) et on ajoute un comparateur qui délivre le signal d'erreur ε afin de réaliser la boucle d'asservissement.

Dans un but de comparaison, on dédouble tout le schéma : une version analogique (sortie sa), et une version numérique (sortie sn), les 2 ayant la même entrée ve.

Le signal appliqué en ve est un échelon unitaire.



Le test final et la comparaison analogique / numérique



Réponse à l'échelon unitaire des 2 asservissements : les courbes sont entièrement confondues

Interprétation

A $t = 1$ s, le signal d'entrée passe à 1 V, puis revient à 0 V à $t = 8$ s.

L'allure des courbes montre que l'asservissement est fonctionnel et performant : pas d'erreur statique (la valeur finale = 1 V), faible dépassement (inférieur à 5 %).

Les 2 courbes sont superposées, ce qui montre que le PID analogique peut être remplacé par son équivalent numérique, sans modification du comportement du système bouclé.

Le processus $H(p)$ étant du type passe bas d'ordre 3, la discrétisation présente en sortie du PID n'est plus visible en sortie de la boucle. La sortie sn est rendue analogique.

Pour information, le réglage de ce PID permet un temps de réponse à 5 % de 1,16 s du système bouclé.

Conclusion

Ce document montre que Pspice (ici, version d'évaluation) est un outil qui permet d'analyser les traitements numériques, que ce soit sur les réponses en fréquence ou les réponses temporelles. Il existe des outils dédiés numériques, plus adaptés, mais Pspice permet de faire des simulations mixtes : circuit numérique et circuit analogique, comme l'exemple 4 de ce document.

Dans tous les cas, c'est une simulation en partie comportementale, car il n'est pas traité le côté « hard » :

- des convertisseurs : nombre de bits (et donc taille des données), temps de conversion,
- des échantillonneurs bloqueurs (temps d'établissement),
- du bloc retard (basculé D, ou autre).

Néanmoins, les réponses obtenues sont très satisfaisantes pour la vérification, compréhension d'un schéma contenant un traitement numérique.

Annexe 1 les méthodes de **transposition**

Il faut tout d'abord disposer de la transmittance sous forme de Laplace du traitement analogique équivalent $T(p)$, pour déterminer son équivalent numérique $T(z)$. La période d'échantillonnage du traitement numérique est T , ce qui revient à écrire que la fréquence d'échantillonnage est $1/T$.

Il en existe 3 connues : méthode du rectangle avant, méthode du rectangle arrière, méthode du trapèze (ou Tustin, ou bilinéaire). La plus usuelle est la méthode de Tustin : la variable p est remplacée par $\frac{2(z-1)}{T(z+1)}$.

Annexe 2 les méthodes de **l'invariant temporel**

Le traitement analogique reçoit un signal d'entrée $v_e(t)$ et délivre en sortie $v_s(t)$

Il en existe 2 connues :

- La méthode de **l'invariant impulsif** : la sortie $v_{s(NT)}$ passe par les mêmes points que $v_s(t)$ à chaque instant d'échantillonnage, lors d'une entrée **impulsion** en consigne $v_e(t)$.

- La méthode de **l'invariant indiciel** : la sortie $v_{s(NT)}$ passe par les mêmes points que $v_s(t)$ à chaque instant d'échantillonnage, lors d'une entrée **échelon** en consigne $v_e(t)$.

Annexe 3 **Extrait des tables des transformées de Laplace, des transformées en Z.**

1^{ère} colonne : un signal à temps continu $x(t)$,

2^{ème} colonne : sa transformée de Laplace, $X(p) = \int_{t=0}^{\infty} x(t) \cdot e^{-pt} \cdot dt$ ($X(p)$ n'est pas une transmittance !)

3^{ème} colonne : sa transformée en Z, $X(z)$, de la séquence $\{x(n \cdot T)\}$, déduite de $x(t)$. $X(z) = \sum_{n=-\infty}^{\infty} x(n \cdot T) \cdot z^{-n}$

Par facilité, on écrit : $X(z) = Z\{x(t)\} = Z\{X(p)\}$.

Mode d'emploi (le côté pratique de ce tableau cache quelques pièges) :

On passe de $x(t)$ à $X(p)$: transformée de Laplace.

On passe de $X(p)$ à $x(t)$: transformée de Laplace inverse.

On passe de $x(t)$ à $X(z)$: transformée en Z.

On ne passe pas de $X(z)$ à $x(t)$: il y a une infinité de $x(t)$ qui ont le même $X(z)$. Il suffit qu'ils aient la même valeur aux instants d'échantillonnage.

On ne passe pas de $X(p)$ à $X(z)$, ni réciproquement.

<i>Signal $x(t)$</i>	Transformée de Laplace : $X(p)$	Transformée en Z : $X(z)$
$\delta(t)$, impulsion de Dirac	1	
1, si $t = 0$, 0 ailleurs.		1
$u(t)$, fonction de Heavyside : 1, si $t \geq 0$, 0 ailleurs.	$\frac{1}{p}$	$\frac{z}{z-1}$
$t u(t)$	$\frac{1}{p^2}$	$\frac{T z}{[z-1]^2}$
$\frac{t^2}{2} u(t)$	$\frac{1}{p^3}$	$\frac{T^2 z [z+1]}{2 [z-1]^3}$
$e^{-a t} u(t)$	$\frac{1}{p+a}$	$\frac{z}{z - e^{-aT}}$

On ne présente que quelques lignes d'un tableau usuellement plus complet. Pour l'exemple 2 page 3 de ce document, il faut utiliser les lignes 3 et 6.

Annexe 4 : fichiers.cir

```
Filtrage numérique
* Fichier exemple_1.cir
```

```
.param a = 0.13793
.param b = -0.13793
.param c = 0.72413
.param tau = 50e-6
.param T = 16e-6
```

```
Vin 1 0 AC=1
Xampli1 1 2 MULTa
Xdelay1 1 3 RETARD
Xampli2 3 4 MULTb
Xadd1 2 4 5 SOUS
Xdelay2 8 7 RETARD
Xampli3 7 6 MULTc
Xadd2 5 6 8 SOMM
```

```
.subckt MULTa in out
Emul out 0 in 0 {a}
.ends
```

```
.subckt MULTb in out
Emul out 0 in 0 {b}
.ends
```

```
.subckt MULTc in out
Emul out 0 in 0 {c}
.ends
```

```
.subckt SOMM in1 in2 out
Esum out 0 value={V(in1)+V(in2)}
.ends
```

```
.subckt SOUS in1 in2 out
Esou out 0 value={V(in1)-V(in2)}
.ends
```

```
.subckt RETARD in out
Ez out 0 LAPLACE {V(in)} {exp(-s*T)}
Raux out 0 1G
.ends
```

```
Ea 9 0 LAPLACE {V(1)} {1/(1+{tau}*s)}
.AC dec 10000 1 100k
.PROBE
.END
```

Nestliste, prête à simuler

```
Filtrage numérique
* Fichier exemple_2.cir
```

```
.param a = 12
.param b = 11.5576
.param c = 0.7788
.param T = 20e-6
```

```
Vin 1 0 PULSE (0 1 0 10n 10n 500u 1m)
```

```
.model INT Vswitch (RON=1m ROFF=1G VON=1 VOFF=0)
Sint 1 10 9 0 INT
```

```
Vech 9 0 PULSE (0 2 0 2n 2n 10n 20u)
```

```
Xampli1 10 2 MULTa
Xdelay1 10 3 RETARD
Xampli2 3 4 MULTb
Xadd1 2 4 5 SOUS
Xdelay2 8 7 RETARD
Xampli3 7 6 MULTc
Xadd2 5 6 8 SOMM
```

```
.subckt MULTa in out
Emul out 0 in 0 {a}
.ends
```

```
.subckt MULTb in out
Emul out 0 in 0 {b}
.ends
```

```
.subckt MULTc in out
Emul out 0 in 0 {c}
.ends
```

```
.subckt SOMM in1 in2 out
Esum out 0 value={V(in1)+V(in2)}
.ends
```

```
.subckt SOUS in1 in2 out
Esou out 0 value={V(in1)-V(in2)}
.ends
```

```
.subckt RETARD in out
Ez out 0 LAPLACE {V(in)} {exp(-s*T)}
.ends
```

```
.TRAN 0.1u 0.5m 0 0.1u
.PROBE
.END
```

Nestliste, prête à simuler

```
Filtrage numérique
* Fichier exemple_3.cir

.param T = 1e-3
.param f0 = 100
.param tau = {1/(2*pi*f0)}
.param a = {T*T + 4*tau*tau}
.param b = {2*T*T - 8*tau*tau}
.param c = {T*T + 4*tau*tau}
.param d = {T*T + 4*tau*T + 4*tau*tau}
.param e = {2*T*T - 8*tau*tau}
.param f = {T*T - 4*tau*T + 4*tau*tau}

Vin 1 0 AC=1
Xampli1 1 2 MULT1
Xdelay1 1 3 RETARD
Xampli2 3 4 MULT2
Xdelay2 3 5 RETARD
Xampli3 5 6 MULT3

Xadd 2 4 6 8 10 Sn SOMM
Xdelay3 Sn 7 RETARD
Xampli4 7 8 MULT4

Xdelay4 7 9 RETARD
Xampli5 9 10 MULT5

.subckt MULT1 in out
Emul out 0 in 0 {a/d}
.ends

.subckt MULT2 in out
Emul out 0 in 0 {b/d}
.ends

.subckt MULT3 in out
Emul out 0 in 0 {c/d}
.ends

.subckt MULT4 in out
Emul out 0 in 0 {e/d}
.ends

.subckt MULT5 in out
Emul out 0 in 0 {f/d}
.ends

.subckt SOMM in1 in2 in3 in4 in5 out
Esum out 0 value={V(in1)+V(in2)+V(in3)-V(in4)-V(in5)}
.ends

.subckt RETARD in out
Ez out 0 LAPLACE {V(in)} {exp(-s*T)}
.ends

Ea Sa 0 LAPLACE {V(1)} {(1+(s*tau)*s*tau)} / ((1+tau*s)*(1+tau*s))
.AC lin 10000 10 1k
.PROBE
.END
```

Netliste, prête à simuler

```
PID numérique
* Fichier KI.cir

.param Ki=0.27
.param T=0.01

Vin Ve 0 PWL (0,0 10u,1) ; PWL (0,0 0.2,0.2)
.model INT Vswitch (RON=1m ROFF=1G VON=1 VOFF=0)
Sint Ve epsn clk 0 INT
Chold epsn 0 10n

Vech clk 0 PULSE (0 2 0u 10u 10u 1m {T})

Xint epsn sin KI

.SUBCKT KI eps SI
Xampli1 eps 2 MULT4
Xdelay1 eps 3 RETARD
Xampli2 3 4 MULT4
Xadd1 2 4 5 SOMM2
Xdelay2 SI 7 RETARD
Xampli3 7 6 MULT5
Xadd2 5 6 SI SOMM2
.ENDS

.subckt SOMM2 in1 in2 out
Esum out 0 value={V(in1)+V(in2)}
.ends

.subckt MULT4 in out
Emul out 0 in 0 {Ki*T/2}
.ends

.subckt MULT5 in out
Emul out 0 in 0 0.99999
.ends

.subckt RETARD in out
Ez out 0 LAPLACE {V(in)} {exp(-s*T)}
.ends

* transmittance analogique témoin
EKi sia 0 LAPLACE {V(Ve)} {{Ki}/s}

.TRAN 10u 200m 0 10u
.PROBE
.END
```

Ne liste, prête à simuler

```
PID numérique
* Fichier KD.cir

.param Kd=1
.param tau=0.05
.param T=0.01

Ve Ve 0 PULSE (0 1 0 10u 10u 5 10)

Vech clk 0 PULSE (0 2 0u 10u 10u 1m {T})
.model INT Vswitch (RON=1m ROFF=1G VON=1 VOFF=0)
Sint Ve epsn clk 0 INT
Chold epsn 0 10n

Xint epsn sdn KD

.subckt MULT1 in out
Emul out 0 in 0 {2*Kd/(2*tau+T)}
.ends

.subckt MULT2 in out
Emul out 0 in 0 {-2*Kd/(2*tau+T)}
.ends

.subckt MULT3 in out
Emul out 0 in 0 {-(T-2*tau)/(2*tau+T)}
.ends

.SUBCKT KD eps Sdn
Xampli1 eps 2 MULT1
Xdelay1 eps 3 RETARD
Xampli2 3 4 MULT2
Xadd1 2 4 5 SOMM2
Xdelay2 Sdn 7 RETARD
Xampli3 7 6 MULT3
Xadd2 5 6 Sdn SOMM2
.ENDS

.subckt SOMM2 in1 in2 out
Esum out 0 value={V(in1)+V(in2)}
.ends

.subckt RETARD in out
Ez out 0 LAPLACE {V(in)} {exp(-s*T)}
.ends

* transmittance analogique témoin
Ea sda 0 LAPLACE {V(Ve)} {{Kd} *s/(1+{tau}*s)}

.TRAN 10u 300m 0 10u
.PROBE
.END
```

Netliste, prête à simuler


```

PID numérique
* Fichier PID.cir

.param Kd=1
.param tau=0.05
.param Ki=0.27 ; 10 pour .TRAN
.param Kp=0.55
.param T=0.01

Ve ve 0 PULSE (0 1 0 10u 10u 5 10) AC=1

* version analogique
EKi sia 0 LAPLACE {V(ve)} {{Ki}/s}
EKd sda 0 LAPLACE {V(ve)} {{Kd}*s/(1+{tau}*s)}
Xadd3a sia sda ve sdi SOMM3
XKpa sdi spa MULT6

.subckt SOUS in1 in2 out
Esum out 0 value={V(in1)-V(in2)}
.ends

.subckt SOMM3 in1 in2 in3 out
Esum out 0 value={V(in1)+V(in2)+V(in3)}
.ends

.subckt MULT6 in out
Emul out 0 in 0 {Kp}
.ends

* version numérique
Vech clk 0 PULSE (0 2 0u 10u 10u 1m {T}) DC=2
.model INT Vswitch (RON=1m ROFF=1G VON=1 VOFF=0)

Sint ve epsn clk 0 INT
Chold epsn 0 10n
Xint ve sin KI
Xder ve sdn KD
Xadd3 sin sdn ve sdi SOMM3
XKp sdi spn MULT6

.subckt MULT1 in out
Emul out 0 in 0 {2*Kd/(2*tau+T)}
.ends

.subckt MULT2 in out
Emul out 0 in 0 {-2*Kd/(2*tau+T)}
.ends

.subckt MULT3 in out
Emul out 0 in 0 {-(T-2*tau)/(2*tau+T)}
.ends

.SUBCKT KD eps Sd
Xampli1 eps 2 MULT1
Xdelay1 eps 3 RETARD
Xampli2 3 4 MULT2
Xadd1 2 4 5 SOMM2
Xdelay2 Sd 7 RETARD
Xampli3 7 6 MULT3
Xadd2 5 6 Sd SOMM2
.ENDS

.subckt SOMM2 in1 in2 out
Esum out 0 value={V(in1)+V(in2)}
.ends

.subckt MULT4 in out
Emul out 0 in 0 {Ki*T/2}
.ends

.subckt MULT5 in out
Emul out 0 in 0 0.99999
.ends

.SUBCKT KI eps SI
Xampli1 eps 2 MULT4
Xdelay1 eps 3 RETARD
Xampli2 3 4 MULT4
Xadd1 2 4 5 SOMM2
Xdelay2 SI 6 RETARD
Xampli3 6 7 MULT5
Xadd2 5 7 SI SOMM2
.ENDS

.subckt RETARD in out
Ez out 0 LAPLACE {V(in)} {exp(-s*{T})}
.ends

.TRAN 10u 1 0 10u
*.AC dec 1000 1m 200
.PROBE
.END

```

```

asservissement complet
* Fichier asservissement_PID.cir

.param Kd=1
.param tau=0.05
.param Ki=0.27
.param Kp=0.55
.param T=0.01 ; Fclk = 100 Hz

Ve ve 0 PULSE (0 1 1 10u 10u 7 14)

* version analogique
Xcompa ve sa epsa SOUS
EKi sia 0 LAPLACE {V(epsa)} {{Ki}/s}
EKd sda 0 LAPLACE {V(epsa)} {{Kd}*s / (1+{tau}*s)}
Xadd3a sia sda epsa sdia SOMM3
XKpa sdia spa MULT6
EHa sa 0 LAPLACE {V(spa)} {10/(s*s*s + 4*s*s + 4*s + 1)}

.subckt SOUS in1 in2 out
Esum out 0 value={V(in1)-V(in2)}
.ends

.subckt SOMM3 in1 in2 in3 out
Esum out 0 value={V(in1)+V(in2)+V(in3)}
.ends

.subckt MULT6 in out
Emul out 0 in 0 {Kp}
.ends

* version numérique
Vech clk 0 PULSE (0 2 0u 10u 10u 1m {T})
.model INT Vswitch (RON=1m ROFF=1G VON=1 VOFF=0)
Sint ep epsn clk 0 INT
Chold epsn 0 10n

Xcompn ve sn ep SOUS
Xint epsn sin KI
Xder epsn sdn KD
Xadd3 sin sdn epsn sdin SOMM3
XKp sdin spn MULT6
EH sn 0 LAPLACE {V(spn)} {10/(s*s*s + 4*s*s + 4*s + 1)}

.subckt MULT1 in out
Emul out 0 in 0 {2*Kd/(2*tau+T)}
.ends

.subckt MULT2 in out
Emul out 0 in 0 {-2*Kd/(2*tau+T)}
.ends

.subckt MULT3 in out
Emul out 0 in 0 {-(T-2*tau)/(2*tau+T)}
.ends

.SUBCKT KD eps Sd
Xampli1 eps 2 MULT1
Xdelay1 eps 3 RETARD
Xampli2 3 4 MULT2
Xadd1 2 4 5 SOMM2
Xdelay2 Sd 6 RETARD
Xampli3 6 7 MULT3
Xadd2 5 7 Sd SOMM2
.ENDS

.subckt SOMM2 in1 in2 out
Esum out 0 value={V(in1)+V(in2)}
.ends

.subckt MULT4 in out
Emul out 0 in 0 {Ki*T/2}
.ends

.subckt MULT5 in out
Emul out 0 in 0 0.99999
.ends

.SUBCKT KI eps SI
Xampli1 eps 2 MULT4
Xdelay1 eps 3 RETARD
Xampli2 3 4 MULT4
Xadd1 2 4 5 SOMM2
Xdelay2 SI 6 RETARD
Xampli3 6 7 MULT5
Xadd2 5 7 SI SOMM2
.ENDS

.subckt RETARD in out
Ez out 0 LAPLACE {V(in)} {exp(-s*{T})}
.ends

.TRAN 100u 15 0 100u
.PROBE
.END

```

articles 1 à 43 : sur le livre

Tableau récapitulatif des articles PDF disponibles sur ce site

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	présentation	document PDF
44	Exemples basiques et des exercices...	présentation	document PDF
45	Un exemple de circuit passif	présentation	document PDF
46	Un oscillateur Colpitts	présentation	document PDF
47	Compensation en fréquence des amplificateurs opérationnels	présentation	document PDF
48	Un amplificateur à transistors bipolaires	présentation	document PDF
49	Une bascule D Flip Flop CMOS	présentation	document PDF
50	Une porte XOR à transistors MOS	présentation	document PDF
51	Un VCO à 12 transistors MOS	présentation	document PDF
52	Une PLL à moins de 20 transistors MOS	présentation	document PDF
53	Un oscillateur à résistance négative	présentation	document PDF
54	Une charge électronique	présentation	document PDF
55	Un amplificateur en classe C	présentation	document PDF
56	Le monostable 74 123	présentation	document PDF
57	Un amplificateur en classe D	présentation	document PDF
58	Le transformateur en linéaire	présentation	document PDF
59	La loi d'ohm thermique	présentation	document PDF
60	Le transformateur en non linéaire	présentation	document PDF
61	Robustesse d'un oscillateur en anneau	présentation	document PDF
62	Une alimentation stabilisée	présentation	document PDF
63	Modélisation d'un haut-parleur	présentation	document PDF
64	Un synthétiseur de fréquence	présentation	document PDF
65	Un ampli audio de Sparkfun	présentation	document PDF
66	Simulation logique et analogique	présentation	document PDF
67	Un oscillateur à relaxation	présentation	document PDF
68	Lecteur de TAG RFID 125 kHz	présentation	document PDF
69	Diagramme de l'œil avec Pspice	présentation	document PDF
70	Un amplificateur hautes fréquences	présentation	document PDF
71	Une bizarrerie enfin expliquée...	présentation	document PDF
72	Comprendre le paramétrage de la FFT	présentation	document PDF
73	La relation de Bennett	présentation	document PDF
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	présentation	document PDF
75	Une horloge biphase sans recouvrement	présentation	document PDF
76	Quelques simulations sur la diode	présentation	document PDF
77	Un ampli classe A, avec transformateur de sortie	présentation	document PDF
78	Des stimuli pour PSpice	présentation	document PDF
79	Simuler le TL431 : zener ajustable	présentation	document PDF
80	Un ADC flash	présentation	document PDF
81	Une chaîne d'acquisition : S&H, ADC, DAC	présentation	document PDF
82	Un amplificateur 50 MHz	présentation	document PDF
83	Un dérivateur non inverseur	présentation	document PDF
84	Un amplificateur bipolaire avec push pull CMOS	présentation	document PDF
85	Rôle des répéteurs logiques dans un circuit intégré	présentation	document PDF
86	Un driver logique CMOS pour charge 50 ohms	présentation	document PDF
87	Des triggers de Schmitt et des applications	présentation	document PDF
88	Un filtre gaussien analogique	présentation	document PDF
89	Un générateur de bruit rose	présentation	document PDF

